Exercice - applications du calcul intégral

Jean-Michel Sarlat
29 avril 2003

Ce document a été généré par la ***fabrique de Syracuse*** *à partir d'un fichier texte contenant des spécifications de contenu et des commandes d'interfaçage avec* ***Maxima****. Le traitement de ce fichier a donné lieu, dans le même temps que ce document, à la création d'une page* html.
URL : http ://melusine.eu.org/syracuse/maxima/

1 Énoncé

1/ Calculer l'aire de la figure limitée par la parabole d'équation $y = 4x - x^2$ et l'axe des abscisses.

2/ Calculer l'aire du segment de la parabole d'équation $y = x^2$ limité par la droite d'équation $y = 3 - 2x$.

3/ Calculer la longueur de l'arc de la courbe d'équation $y = \ln(x)$ entre les points d'abscisses $x = \sqrt{3}$ et $x = \sqrt{8}$.

4/ Calculer la longueur de l'arc de la courbe d'équation $y = e^x$ entre les points d'abscisses $x = 0$ et $x = 1$.

5/ Calculer le volume du corps engendré par la rotation autour de l'axe Ox de la figure limitée par l'axe Ox et et la parabole d'équation $y = ax - x^2$ $(a > 0)$.

6/ Calculer le volume du tore de rayon extérieur R et de rayon intérieur r.

7/ Déterminer le centre de gravité d'un secteur circulaire de rayon a et d'angle au centre 2α.

2 Corrigé

maxima >>

```
(C2) load("integration.mc")$
```

1/ Les abscisses des points d'intersection de la parabole avec l'axe sont *visibles* : 0 et 4. La parabole est au dessus de l'axe entre ces points.

```
(C3) integre(4*x-x^2,x,0,4);
```

(D3)
$$\int_0^4 4\,x - x^2\,dx = \frac{32}{3}$$

2/ On commence par rechercher les abscisses des points d'intersection des deux courbes.

```
(C4) solve(x^2=3-2*x,x);
```

(D4) $$[x = -3, x = 1]$$

Il y en a deux, on intègre la différence des ordonnées entre ces deux abscisses.

```
(C5) integre(abs(x^2-(3-2*x)),x,rhs(part(%,1)),rhs(part(%,2)));
```

(D5) $$\int_{-3}^{1} \left|x^2 + 2\,x - 3\right|\,dx = \frac{32}{3}$$

Tiens, c'est le même résultat qu'à la question précédente.
3/ Nous allons commencer par définir la longueur d'un arc de courbe à l'aide d'une intégrale.

```
(C6) s(f,a,b):=integrate(sqrt(1+diff(f(x),x)^2),x,a,b);
```

(D6) $$s\left(f, a, b\right) = \int_{a}^{b} \sqrt{\left(\frac{d}{d\,x}\,f\left(x\right)\right)^2 + 1}\,dx$$

```
(C7) s(log,sqrt(3),sqrt(8)),logcontract;
```

(D7) $$\frac{\ln\left(\frac{3}{2}\right) + 2}{2}$$

Le calcul est suivi de l'appel à `logcontract` qui simplifie les logarithmes dispersés dans l'expression.
4/ On réutilise la fonction **s** précédente.

```
(C8) s(exp,0,1);
```

(D8) $$-\operatorname{Argsh}\left(e^{-1}\right) + \operatorname{Argsh}\left(1\right) + \sqrt{e^2 + 1} - \sqrt{2}$$

La fonction Argsh pouvant s'exprimer à l'aide du logarithme, nous allons procéder à une substitution.

```
(C9) subst(lambda([x],log(x+sqrt(1+x^2))),ASINH,%),logcontract;
```

(D9) $$\ln\left(\frac{\left(\sqrt{2} + 1\right)\,e}{\sqrt{e^2 + 1} + 1}\right) + \sqrt{e^2 + 1} - \sqrt{2}$$

On a donc substitué $x \mapsto \ln\left(x + \sqrt{1 + x^2}\right)$ à Argsh, sans *nommer* la première fonction, en utilisant le préfixe `lambda`.
5/ Le volume est à considérer comme un empilement de disques de rayon $ax - x^2$ et d'épaisseur dx, pour x variant de 0 à a.

```
(C10) integre(%pi*(a*x-x^2)^2,x,0,a);
```

(D10) $$\pi \int_0^a \left(a\,x - x^2\right)^2\,dx = \frac{\pi\,a^5}{30}$$

6/ Le tore est à considérer comme un empilement de couronnes circulaires de rayons $R + \sqrt{r^2 - z^2}$ et $R - \sqrt{r^2 - z^2}$, d'épaisseur dz pour z variant de $-r$ à $+r$.

```
(C11) assume(r>0)$

(C12) integre(%pi*((R+sqrt(r^2-z^2))^2-(R-sqrt(r^2-z^2))^2),z,-r,r);
```

(D12) $$\pi \int_{-r}^{r} \left(\sqrt{r^2 - z^2} + R\right)^2 - \left(R - \sqrt{r^2 - z^2}\right)^2\,dz = 2\,\pi^2\,R\,r^2$$

7/ On prend l'axe de symétrie du secteur angulaire comme axe des abscisses. Ainsi l'ordonnée du centre de gravité de la plaque (supposée homogène) est nulle. Reste à calculer l'abscisse, pour cela nous allons utiliser les coordonnées polaires.

```
(C13) assume(alpha>0)$

(C14) N:integrate(integrate(r*cos(theta)*r,r,0,a),theta,-alpha,alpha);
```

(D14) $$\frac{2\,a^3\,\sin\alpha}{3}$$

```
(C15) D:integrate(integrate(r,r,0,a),theta,-alpha,alpha);
```

(D15) $$a^2\,\alpha$$

Il reste à faire le quotient des quantités précédentes (aire pondérée par l'abscisse et aire du secteur angulaire) pour obtenir l'abscisse du centre de gravité.

```
(C16) N/D;
```

(D16) $$\frac{2\,a\,\sin\alpha}{3\,\alpha}$$

On fait tendre α vers 0 pour vérifier si l'on obtient bien alors la position du centre de gravité sur la médiane d'un triangle mesurée à partir du sommet.

```
(C17) limit(%,alpha,0);
```

(D17) $$\frac{2\,a}{3}$$

C'est bon !